
===========Hibert_AllPass_Analog================

*                              ___
*                        _____| b2|_____________________
*                       |     |___|     ___             |
*                       |          ____| b1|__________  |
*                       |Highpass |    |___|          | |
* U(S)        ___   ..  |         | .                 V_V
*     ___    / _ \  y(t)|  _____  | y(t) _____ y(t)  / _ \
*  _\| k |_\/ \   \_____|_| 1/s |_|_____| 1/s |____\/ \   \___\ Y(S)
*   /|___| /\ /_  /       |_____|   |   |_____||   /\ /_  /   /
*            \___/                  |          |     \___/
*             ^ ^    _____ Bandpass |          |
*             | |___| -a2 |_________|          |  b0 = 1  
*             |     |_____|  _____     Lowpass |
*             |_____________| -a1 |____________|   w0 = sqrt(a1)
*                           |_____|                Q  = w0/a2
*
*         Y(S)/U(S)  =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*         Complex_Poles  1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)

An old method to building wide bandwidth 90 degree phase
shifters was to use two 6 pole allpass filters. 

VFC1      FC1    0      DC     1
VK        K      0      DC     1
VA1       A1     0      DC     1
VA2       A2     0      DC     1
VB0       B0     0      DC     1
VB1       B1     0      DC     -1
VB2       B2     0      DC     1
XStateS1  VIN    FC1    K      A1  A2  B0  B1  B2  VOUT1  HP   BP   LP   StateVS

A DC controlled subcircuit has been built to do the ALL Pass.

VFC1      FC1    0      DC     1
VFC2      FC2    0      DC     3
BVFC3     FC3    0      V      = V(FC2)*V(FC2) 
XStateS1  VIN    FC1    K   A1  A2  B0  B1  B2  VOUT1  HP   BP   LP   StateVS
XStateS2  VIN    FC2    K   A1  A2  B0  B1  B2  VOUT2  HP2  BP2  LP2  StateVS
XStateS3  VIN    FC3    K   A1  A2  B0  B1  B2  VOUT3  HP3  BP3  LP3  StateVS 

A Frequency spacing of 3 will be use for each All Pass.



VFC4      FC4    0      DC     1.37  
BVFC5     FC5    0      V      = V(FC4)*V(FC2) 
BVFC6     FC6    0      V      = V(FC4)*V(FC3) 
XStateS4  VIN    FC4    K      A1  A2  B0  B1  B2  VOUT4  HP4  BP4  LP4  StateVS
XStateS5  VIN    FC5    K      A1  A2  B0  B1  B2  VOUT5  HP5  BP5  LP5  StateVS
XStateS6  VIN    FC6    K      A1  A2  B0  B1  B2  VOUT6  HP6  BP6  LP6  StateVS

An additional three stages are built up which have 1.37 additional frequency 
offset. 

The idea is to maintain 90 degrees of phase shift between the first 6 pole all pass 
to the second 6 pole all pass. Adding more circuit elements will make things better. 

The closer the phase offset between the two 6 pole AllPass filters can be maintained
at 90 dgrees, the better the sideband cancelation can take place to produce a 
single sideband output. 

===========Full_Netlist_For_Copy_Paste=======================
Hibert_AllPass
*                              ___
*                        _____| b2|_____________________
*                       |     |___|     ___             |
*                       |          ____| b1|__________  |
*                       |Highpass |    |___|          | |
* U(S)        ___   ..  |         | .                 V_V
*     ___    / _ \  y(t)|  _____  | y(t) _____ y(t)  / _ \
*  _\| k |_\/ \   \_____|_| 1/s |_|_____| 1/s |____\/ \   \___\ Y(S)
*   /|___| /\ /_  /       |_____|   |   |_____||   /\ /_  /   /
*            \___/                  |          |     \___/
*             ^ ^    _____ Bandpass |          |
*             | |___| -a2 |_________|          |  b0 = 1  
*             |     |_____|  _____     Lowpass |
*             |_____________| -a1 |____________|   w0 = sqrt(a1)
*                           |_____|                Q  = w0/a2
*
*         Y(S)/U(S)  =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*         Complex_Poles  1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)
*
*.OPTIONS GMIN=1p METHOD=TRAP ABSTOL=1u TEMP=27 srcsteps = 1 
*.OPTIONS RELTOL=.001 ABSTOL=1n VNTOL=1n ITL4=500 
.OPTIONS GMIN=1p  srcsteps = 1 ITL1=4000 gminsteps = 1

*=========Create_Signal==================
VIN       VIN    0      DC     0   AC   1
VFC1      FC1    0      DC     1
VFC2      FC2    0      DC     3
BVFC3     FC3    0      V      = V(FC2)*V(FC2) 

VFC4      FC4    0      DC     1.37  
BVFC5     FC5    0      V      = V(FC4)*V(FC2) 
BVFC6     FC6    0      V      = V(FC4)*V(FC3) 



VK        K      0      DC     1
VA1       A1     0      DC     1
VA2       A2     0      DC     1
VB0       B0     0      DC     1
VB1       B1     0      DC     -1
VB2       B2     0      DC     1

XStateS1  VIN    FC1    K      A1  A2  B0  B1  B2  VOUT1  HP1  BP1  LP1  StateVS
XStateS2  VIN    FC2    K      A1  A2  B0  B1  B2  VOUT2  HP2  BP2  LP2  StateVS
XStateS3  VIN    FC3    K      A1  A2  B0  B1  B2  VOUT3  HP3  BP3  LP3  StateVS

XStateS4  VIN    FC4    K      A1  A2  B0  B1  B2  VOUT4  HP4  BP4  LP4  StateVS
XStateS5  VIN    FC5    K      A1  A2  B0  B1  B2  VOUT5  HP5  BP5  LP5  StateVS
XStateS6  VIN    FC6    K      A1  A2  B0  B1  B2  VOUT6  HP6  BP6  LP6  StateVS

.control
set       pensize = 2
*AC       DECLin NUMDEC FSTART FSTOP  
ac        dec    100    .3     30    
let pht1  = unwrap(ph(vout1)) + unwrap(ph(vout2))  + unwrap(ph(vout3))
let pht2  = unwrap(ph(vout4)) + unwrap(ph(vout5))  + unwrap(ph(vout6)) 

plot      db(vout2) unwrap(ph(vout2))                                                  title AllPassOut 
plot      db(vout1) unwrap(ph(vout1)) unwrap(ph(vout2)) unwrap(ph(vout3)) pht1         title AllPassOut1 
plot      db(vout4) unwrap(ph(vout4)) unwrap(ph(vout5)) unwrap(ph(vout6)) pht2         title AllPassOut2 

plot      pht1  pht2          title Phase1_Phase2  
plot      pht1- pht2          title Delta_Phase  ylimit -100 0

.endc
*=========StateVariable_Cell_S===========================================
*
*
* ___   |\  K      HP  |\  b2 gmb2                       ___
*| IN|__| \/\/\________| \/\/\__________________________|OUT|
*|___|  | /\/\/     |  | /\/\/                        | |___|
*       |/  gmin    |  |/       |\  b1 gmb1           |
*                   |         __| \/\/\_______________|
*                   |         | | /\/\/               |
*                   |  Fscale | |/   Fscale           |
*          R1 =1    |  __|__  |      __|__    |\  b0  |     R2 =1
*     _/\  /\  /\___|_| 1/s |_|_____| 1/s |___| \/\/\_|_/\  /\  /\_
*    _|_ \/  \/   |   |_____|   |   |_____||  | /\/\/     \/  \/  _|_
*    ///          | s1       /| |      s2  |  |/  gmb0            ///
*                 |_____/\/\/ |_|BP        |
*                 |     \/\/\ |gma1        |
*                 |     -a2  \|         /| |LP
*                 |________________/\/\/ |_|
*                                  \/\/\ |gma2
*                                  -a1  \|
.SUBCKT   StateVS VIN    FC   K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP 
 R1        HP     0      1
 R2        OUT    0      1
 Bgmin     HP     0      I  = -V(VIN)*V(K)*1
 Bgma1     HP     0      I  =  V(LP)*V(A1)
 Bgma2     HP     0      I  =  (V(BP))*V(A2)
 Bgmb0     OUT    0      I  = -V(LP)*V(B0)
 Bgmb1     OUT    0      I  = -V(BP)*V(B1)
 Bgmb2     OUT    0      I  = -V(HP)*V(B2)
 XS1block  HP     BP     FC    Sblock
 XS2block  BP     LP     FC    Sblock
 BOUT      VOUT   0      V  =  V(OUT)
.ENDS     StateVS
*=========S_BLOCK===========================================
*                         _______
*         IN   RS=1/FC  _|_      |
* ___   |\              ___      |
*|VIN|__| \_/\  /\  /\___|___|\  |  ___
*|___|  | /   \/  \/   VNEG  | \_|_|OUT|
*       |/Gain=-1        ____| /   |___|
*                       _|_  |/
*   For FC = 1Hz        ///
*   RS = 1 Ohm               CS=1uF/2Pi
*   Xc = 1 Ohm
.SUBCKT   Sblock VIN    OUT    FC  
 Bbuf     IN     0      V =    -V(VIN)
 BRS      IN     VNEG   I =    (V(IN)-V(VNEG))*V(FC)
 Cs       VNEG   OUT    .159      
 BSOUT    OUT    0      V =    -V(VNEG)*3000
.ENDS     Sblock
.end

